

>CURSOR

The Tiny Micro Computer News Service -

CONGRATULATIONS!

You have purchased the most powerful and diverse micro-computer on the home market (and you probably don't know it). Chin up. The Cursor staff is standing in the wings, aching to answer your questions, untangle the mess, and put you on the road to a fulfilling and enjoyable pastime.

Have you reached the point where you would like to do more than move a box around the screen? We are waiting to show you step-by-step how to perform decimal math routines with 16 digit accuracy (far better than any calculator). We'll send you a written "Hangman" program with quality far superior to any of the commercial cartridges. How about a four (4) player game of "Aggravation" or "Othello"? How would you like a program that analyzes your electric bill and tells you how much you pay per month to run each appliance? How about Bio-rythms? Would you like to put five (5) colors on screen at one time? Assembly Language?

Are you interested in adding a Printer? How about a full sized Type-writer Keyboard? Would you like to have a Telephone Data Transfer Modem? Possibly you might like to add more memory. We've done it!! And we'll be thrilled to show you how.

Cursor offers the following services:


1. Monthly Newsletter: All the above named examples have been printed on our pages or are about to be printed (plus much more).
2. Program & Tutorial Exchange: A personal file is established for each subscriber. This allows an individual to send in one of his own programs or tutorials and specifically request the style and format of what he desires in return (70¢ fee charged per transaction to cover postage, copying, filing, etc.).

Each monthly issue contains a minimum of four (4) carefully selected major programs, such as "Othello", "TinyTrek", "Life Synthesis Model", etc. These programs are printed "glitch-free" by using final copy for input - if it runs, we print it. Many issues contain programming contests, with prizes awarded to monthly winners. Winning entries are published in subsequent issues.

Software and Hardware are offered for sale in our "Classified" section. All products are required to meet certain quality criteria, i.e., level of complexity, originality, documentation, capture of interest, etc.

By now, you are asking - "What is the bonanza going to cost me?". Would you be willing to spend 1/3 of the price of a game cartridge? Our fee for all services inclusive is \$9.50 (six month subscription). Please make all checks payable to the undersigned.

Once again, we welcome you to the fold, and are standing by, waiting to serve your needs.



FRED CORNETT
6115 Clybourn #25
No. Hollywood, CA 91606
(213) 763-7701

ISSUE

>CURSOR

The Tiny Micro Computer News Service -

VOLUMN I \$2.00 per Issue JANUARY 1980 \$9.50 Semi-Annually ISSUE 1

AT LAST!!

Cursor's New Years gift to you is getting this issue in your mailbox prior to the 4th of July. Bally has invited the Cursor staff to the "International Winter Consumer Electronics Show" being held in Las Vegas in January. Bally is planning a private show in their Hilton suite. We will tell you all about it in Issue #2.

We have included three programs in this issue, two of which are totally the design (for better or worse) of Fred Cornett (Plastic Puzzle & Life Synthesis Model), the other program is stolen from another publication, and has been rewritten for the Bally (for the purpose of better understanding conversion from one BASIC to another): "Electric Bill Analysis" is a translated version of "POWER" by Karen S. Wolfe (written in North Star BASIC) in the October issue of BYTE. I have had many, many requests for a tutorial regarding the translation of printed programs written in other versions of BASIC.

For those of you desiring the ability to translate other BASIC programs to Bally Tiny BASIC, I can only offer 2 bits of advice:

1. You must fully understand all Tiny BASIC statements before attempting any translation.
2. You must fully understand the BASIC statement you wish to translate.

Since most of us don't have the ability to work interchangeably with other BASIC versions, I would, without reservation, recommend the purchase of "The Basic Handbook" by David A. Lien. This book explains the purpose of most statements utilized by all major versions of BASIC.

ELECTRIC BILL

This program utilizes the "internal calculator", which renders an answer with 16 digit accuracy in the following format: NNNNNNNN.NNNNNNNN

This routine uses 18 string variables @(A), to load a number. The instruction is: \$F@(A),@(B),@(C) Where F=Function (:x-+)

A=Beginning location of number to be manipulated

B=Beginning location of number doing the manipulating

C=Beginning location of Answer

I have bypassed the proper utilization of the internal calculator routine by formatting, instead of leading zero suppression, etc. I did not use overflow or sign indicators as they would not apply. Also, I did not make any attempt to save bytes as I felt it was unnecessary. Enough space is left for considerable expansion for those so desirous.

For those who would like to know more about the internal calculator, I have prepared a complete tutorial with programs and step by step instructions. Merely drop me a line and 75¢ (to cover copying, postage & handling) and I'll send it by return mail. (Not enough space here to explain the process.)

When prompted for \$ Amount of electric bill, please input the Total of all \$ amounts, tax, fees, etc.

NOTE: You may have noticed that the Tiny BASIC statements used in this newsletter differ in format from the statements you are used to using. Do not alter the statements to conform to what you think the statement should look like; such as adding final quotes, etc., the statements herein use a shorthand Tiny Basic which will work!

ELECTRIC BILL ANALYSIS
PROGRAM EXPLANATION

- 1-163 Input Info
- 180 Subtract beginning reading at @(18) from ending reading at @(0) and place answer (KW Hours) at @(54)
Divide amount of bill at @(36) by KW Hours at @(54) and place answer (Price per KW Hour) at @(72)
- 260 Clear registers at: @(0), @(18), @(36), @(54)
- 270 Clear registers at: @(90), @(108), @(126), @(144)
- 350-380 Input info
- 390 Multiply wattage at @(0) by Hours of Appliance Use at @(126) and Place answer (Watt/Hours) at @(36)
Clear @(0), @(18), and forces "1000" at @(0)
- 395 Divide Watt/Hours at @(36) by "1000" at @(0) and place answer (Kilowatt/Hours) at @(18)
- 400 Multiply Kilowatt/Hours at @(18) by Price per KW/Hour at @(72) and place answer at @(54)
- 410-483 Print Answers

NOTE: All listings printed in Cursor utilize the same Special Character Set As your "ARCADE". In other words: What you see is what you type, i.e., Bally Multiplication Sign=Lower Case X on listing (5x4).

AUTOMATIC RUN AFTER LOADING TAPE: When you are ready to save a program on tape; instead of typing :PRINT; LIST
type: NT=1;:PRINT ;LIST ;PRINT ":RETURN ;TV=13;NT=3;RUN
Do not press go until tape is recording. After you press GO, the program will be saved on tape as before with the exception that upon completion of taping, it will print: :RETURN ;TV=13;NT=3;RUN on tape. When you input the program later, it will, upon completion of the load, run the program automatically even if you forgot to turn off the Tape Recorder.

ELECTRIC BILL ANALYSIS SZ=630

```
1 NT=1
10 CLEAR ;FOR A=0TO 143;@(A)=0;NEXT A;BC=10;FC=6
20 PRINT "THIS PROGRAM ANALYZES POWER USAGE AND COST
40 FOR A=1TO 2000;NEXT A
120 GOSUB 500;PRINT "INPUT ENDING METER READING FROM ELECTRIC BILL (XXXX)
122 FOR E=12TO 8STEP -1;@(E)=KP;TV=@(E);NEXT E
140 GOSUB 500;PRINT "INPUT BEGINNING METER";PRINT "READING FROM BILL (XXXX)
142 FOR B=30TO 26STEP -1;@(B)=KP;TV=@(B);NEXT B
160 GOSUB 500;PRINT "INPUT $ AMOUNT OF BILL (XXX.XX)
162 FOR T=46TO 42STEP -1;IF T=43PRINT ".",
163 @(T)=KP;TV=@(T);NEXT T
180 $=@(0),@(18),@(54);$=@(36),@(54),@(72)
260 FOR A=71TO 0STEP -1;@(A)=0;NEXT A
270 FOR A=149TO 90STEP -1;@(A)=0;NEXT A;CLEAR
350 GOSUB 500;PRINT "SELECT AN APPLIANCE AND INPUT WATTAGE (XXX)
360 PRINT ;FOR W=10TO 8STEP -1;@(W)=KP;TV=@(W);NEXT W
370 GOSUB 500;PRINT "INPUT THE ESTIMATED TOTAL HOURS OF APPLIANCE USE
375 PRINT "DURING BILLING PERIOD (XXXX)
380 FOR H=137TO 134STEP -1;@(H)=KP;TV=@(H);NEXT H
390 $x@(0),@(126),@(36);FOR A=35TO 0STEP -1;@(A)=0;NEXT A;@(11)=49
395 $=@(36),@(0),@(18)
400 $x@(18),@(72),@(54)
410 CLEAR
420 PRINT "YOU PAY PER KWHOUR=$0.",
422 FOR C=79TO 76STEP -1;TV=@(C);NEXT C;PRINT
430 PRINT
440 PRINT "APPLIANCE USED ",;FOR K=29TO 24STEP -1;IF K=25PRINT ".",
445 TV=@(K);NEXT K;PRINT ;PRINT "KW HOURS OF POWER
450 PRINT
460 PRINT "BILLING PERIOD COST OF THAT POWER USEAGE=$",
465 FOR D=64TO 60STEP -1;IF D=61PRINT ".",
466 TV=@(D);NEXT D;PRINT
470 PRINT
480 PRINT "COST PER EST. HOUR'S USE =$0.",
482 $=@(54),@(126),@(108)
483 FOR Z=115TO 112STEP -1;TV=@(Z);NEXT Z
490 PRINT " PRESS ANY NUMBER",
495 A=KP;GOTO 260
500 CLEAR ;PRINT "PLEASE PRECEDE WITH ZEROES";PRINT ;RETURN
```

FEBRUARY ISSUE WILL
FEATURE:
Step-by step (idiot proof) instructions on building your own \$20. Telephone Modem.